FORTRAN Coding Standard in PALM	Version 12-Oct-2016 
This document defines a set of FORTRAN specifications, rules, and recommendations for the coding of PALM. It is based on and very similar to specifications given for the ocean dynamics model NEMO[footnoteRef:2]. The purpose is to provide a framework that enables users and developers to easily understand or modify code, or to port it to new computational environments. In addition, it is hoped that adherence to these guidelines will facilitate the incorporation of new packages in PALM. Other work that influenced the development of this standard are Community Climate System Model, Software Developer's Guide[footnoteRef:3], Report on Column Physics Standards[footnoteRef:4], and European Standards For Writing and Documenting Exchangeable FORTRAN 90 Code[footnoteRef:5]. [2: 	 http://www.nemo-ocean.eu/content/download/250/1629/file/coding_rules_OPA9.pdf]  [3: 	 http://www.ccsm.ucar.edu/csm/working_groups/Software/dev_guide/dev_guide/]  [4: 	 http://nsipp.gsfc.nasa.gov/infra/]  [5: 	 http://nsipp.gsfc.nasa.gov/infra/eurorules.html] 


Every code segment submitted to the trunk folder of the PALM repository will be checked by the PALM code supervisors and may be modified in case that the code does not follow the rules which are outlined below.

[bookmark: _GoBack]Current code supervisors are Siegfried Raasch, Björn Maronga, Matthias Sühring and Farah Kanani-Sühring from IMUK-LUH.


Language

· Fortran 2003 standard: The PALM software adheres to the Fortran 2003 language standard and does not rely on any specific language or vendor extensions.
· English language: Owing to the international user community, all naming of variables, modules, functions, and subroutines as well as all comments are to be written in English.
· Physical units: Physical SI-units are used. The temperature is expressed in Kelvin. CGS-units are partly used in the cloud physics module for convenience (historical reasons), e.g. for the dissipation rate. In such cases, conversion factors are required in the equations, and we always give respective comments in front of these equations. 
· Free form source: Free-form source is used. The Fortran 2003 standard allows up to 132 characters. PALM recommendation is to use a self-imposed limit of about 80 in order to enhance readability and make life easier for those with bad eyesight, who wish to make presentation slides of source code, or print source files with two columns per page. We do not recommend a strictly application of this rule: if the readability is improved with more than 80 characters, do not worry about the printing of source files. However, 100 characters should not be exceeded, except in the variable declaration/commenting part of the routines we allow up to 120 characters. Nevertheless, multi-line comments that extend to column 100 would be unacceptable.
· Features to be avoided: In terms of keeping code up to date and easier to maintain the code should always follow the current standards of Fortran and ANSI C. We decide to restrict the languages use to the elements, which are not obsolete or deleted -- even if they are still available with almost all compilers. In particular, the code must be such that during the compiling phase, no warning appears, so that user can be easily alerted of potential bugs when some appear in their configuration.
Examples for Fortran 95 are:
· COMMON blocks - use the declaration part of MODULEs instead.
· EQUIVALENCE - use POINTERs or derived data types instead to form data structures. Please try to avoid this anyway as it is usually a source of bugs.
· Assigned and computed GOTOs - use the CASE construct instead.
· Arithmetic IF statements - use the block IF, ELSE, ELSEIF, ENDIF or SELECT CASE construct instead.
· Labeled DO constructs - use unlabeled ENDDO instead. Nevertheless non-number label can be used for big iterative loop of recurcive algorithm.
· I/O routines END and ERR - use IOSTAT instead (the use is somehow restricted due to compiler implementation dependent error numbering). (This is not yet done throughout the PALM code  action!)
· FORMAT statements: use character parameters or explicit format- specifiers inside the READ or WRITE statement instead. (This is not yet done throughout the PALM code  action!)
· GOTO and CONTINUE statement - use IF, CASE, DO WHILE, EXIT or CYCLE statements or a contained SUBROUTINE instead. If you feel you cannot avoid a GOTO and or CONTINUE statement, then add a clear comment to explain what is going on and why you need to use GOTO.
· PAUSE – just never use it.
· ENTRY statements: a subprogram must only have one entry point.
· RETURN – it is obsolete and so not necessary at the end of program units.
· Fixed source form – use free form instead.
· Avoid functions with side effects. There are good reasons to avoid this. First, the code is easier to understand, if you can rely on the rule that functions don't change their arguments, second, some compilers generate more efficient code for PURE (in Fortran 2003 there are the attributes PURE and ELEMENTAL) functions, because they can store the arguments in different places. This is especially important on massive parallel and on vector machines as well.
· DATA and BLOCK DATA - initialisers in Fortran 2003 give this functionality.


Style and Layout Rules: Generalities

Well thought out and structured code will make all the tasks of debugging, optimisation, parallelisation, etc. far easier and also those of porting, maintenance and adding new functionality. A small extra effort at the start reaps rewards in the long term. One can never underestimate the importance of good software practice when writing code. Not only will this reduce the possibility of bugs being introduced but it may well prolong the lifetime of the code and make it more comprehensible and therefore likely to be more widely used.
Good coding standards are essential when working on larger projects like PALM with more than one person involved, when other people will have to use the code and/or add development to it. We do not have the pretension of having chosen the best style (if it exists!), we have just adopted what we think is good. The key idea here is to be consistent in term of coding rules throughout the code.
On the whole, the main contributor to code-level documentation and readability is not only comments, but rather good programming style. Style includes good program structure, use of straightforward and easily understandable approaches, good variable names, good routine names, use of named constants instead of literals, clear layout, and minimisation of controlflow and data-structure complexity.

When a piece of code is written and you have checked informally that everything looks
reasonable, clean up the leftovers from the development process.

· Check that all parameters are used and that all input and output data is accounted for.
· Check layout and style, i.e. make sure there is white space to clarify the logical structure of the routine, expressions and parameter list.
· Check the documentation, including comments, and make sure that it is up to date and consistent.

This might be an iterative process. Try to understand each line of your code. Avoid the rush to completion and do not compile it until you have convinced yourself that the routine is likely to be correct. Otherwise you might get caught in a hacker-minded attitude leading to hasty, error-prone changes that take more time to debug in the long run.


Documentation – Comments

Documentation consists of putting information both inside and outside the source code. On large, formal projects, most of the documentation is outside the source code. However because the internal documentation is most closely associated with the code, it is the part most likely to remain correct as the code is modified. The thing to be really careful with is that misleading comments are even worse than having no comments at all. So, please report any misleading comments you found. Also try to avoid having superfluous comments such as:

a = a + 1 ! Increment a by one.

This is not adding anything that is not immediately obvious. A person looking through your code should only have to scan your comments to be able to get a good idea of what the code does and where to look for any special activity. A comment should always precede the code it describes or should be put at the end of the same line of code. If necessary spread the comment over several lines – comments that are wrapped by printers on to new lines are not helpful.

Good comments do not just repeat verbally what is happening in the code or just explain it. They clarify its intent. They should explain at a higher level of abstraction what the code or what you are trying to do.

In addition you should comment:

· Everything that gets around an error or an undocumented feature in a language or environment.
· Any violations of good programming style should be justified. This will ensure that any well-intentioned programmer does not break your code by changing the source to implement a “better” style. Comments are especially helpful at the end of long or nested loops. A comment can be used at the end of each loop to show which loop has ended and thus ease the clarity of the code.
· The lines before a control structure, e.g. IF, CASE, loop, etc. are a natural spot to comment and explain what these constructs are about to do.


Naming conventions

Another point, which may seem rather trivial, is the actual naming convention used for the routines. In general, a routine should have a clear, unambiguous name, describing what the routine does. For example:

· Using a verb followed by the object it operates on for a procedure name immediately conveys what the routine is actually doing: e.g. print_report(), read_data(), etc.
· As a function usually returns a value, the function should be named so as to relate some information about the value it is expected to return, e.g. cos(), next_task_id();
· Ambiguous verbs that could cover several meanings should be avoided, e.g. deal_with_input(), handle_calculation();
· Never use names that may clash with the operating system or language intrinsics, e.g. read() or access().

With a little bit of thought you can, from the very start, add to the clarity of the code by choosing your function names, and this is equally applicable to your variables, in a meaningful way and not haphazardly to tickle your fancy.

· Use meaningful English variable and constant names. Choose your variable names so that they are readable, memorable and descriptive. Nonetheless, try to avoid names that are too long, e.g. NumberOfIterationsOfTheMainLoop which is just asking for repetitive strain injury. The use of upper case letter in a variable name is also prohibited. Separate words in variable names by the underline character “_”, e.g. “number_of_iterations”. One letter variable is prohibited, except for the basic flow variables like velocities (u, v, w), humidity (q) or turbulent kinetic energy (e), as well as loop or other counters (e.g. i, j, k).
· Be consistent in your naming convention within a program. When you use an enumerated type, you can ensure that it is immediately clear that members of the same type all belong to the same group by using a common group prefix or suffix, e.g. COLOUR Red, COLOUR Green, COLOUR Blue are all of the type COLOUR. Give boolean variables names that imply True or False. This works fine for DONE, but not for STATUS. When naming constants, name the abstract entity of what the constant represents rather than the number the constant refers to. In general, the degree of formality in naming conventions depends on the number of people working on a code, its size and its expected life span.


Pre-processor:

Where the use of a language pre-processor is required, it will be the C Pre-Processor (CPP). CPP is available on any UNIX platform, and many FORTRAN compilers have the ability to run CPP automatically as part of the compilation process. Nevertheless, use and abuse of CPP to defined numerical and physical option is not recommended. Caution, the #ifdef abbreviate form must not be used. It may have problems with some UNIX scripts. Use the standard #if defined form. Use the standard logical operators “!”, “||” and “&&” in logical expressions, for example

#if ! defined( __parallel ) && ( __mpifh )

instead of

#if .NOT. defined( __parallel ) .AND. ( __mpifh )


Program units:

· Use meaningful English module and subroutine names.
· Always use the same name in lower case for a module file and the inside module. This simplifies the scripts used to compile the system.
· Put one and only one module in a file.
· Always name program units and always use the corresponding END PROGRAM, END SUBROUTINE, END INTERFACE, END MODULE, etc. construct, again specifying the name of the program unit. This helps finding the end of the current program entity.
· Error conditions: When an error condition occurs inside a routine, a message describing what went wrong will be printed. The name of the routine in which the error occurred must be included. Use the message-subroutine which is part of PALM for output of error conditions and any kind of other messages. (should be explained in more detail with all parameters here or better at some place further below)


File format

All the routines must be embedded in a module (not yet systematically done  action). Embedding multiple routines within a single file and/or module is allowed, encouraged in fact, if any of three conditions hold.
1. If routine A and only routine A is in routine B, then the two routines may be included in the same file. This construct has the advantage that inlining A into B is often much easier for compilers if both A and B are in the same file. Practical experience with many compilers has shown that inlining when A and B are in different files often is too complicated for most people to consider worthwhile investigating.
2. The second condition in which it is desirable to put multiple routines in a single file is when they are "CONTAIN"ed in a module for the purpose of providing an implicit interface block. This type of construct is strongly encouraged, as it allows the compiler to perform argument consistency checking across routine boundaries.
3. The final reason to store multiple routines and their data in a single module is that the scope of the data defined in the module can be limited to only the routines which are also in the module. This is accomplished with the "private" clause.
If none of the above conditions hold, it is not acceptable to simply glue together a bunch of functions or subroutines in a single file.

· Module names: Modules MUST be named the same as the file in which they reside. The reason to enforce this as a hard rule has to do with the fact that dependency rules used by "make" programs are based on file names. For example, if routine A "USE"s module B, then "make" must be told of the dependency relation which requires B to be compiled before A. If one can assume that module B resides in file B.o, building a tool to generate this dependency rule (e.g. A.o: B.o) is quite simple. Put another way, it is difficult (to say nothing of CPUintensive) to search an entire source tree to find the file in which module B resides for each routine or module which "USE"s B. Note that by implication multiple modules are not allowed in a single file.
· I/O error conditions: I/O statements which need to check an error condition will use the "iostat=<integer variable>" construct instead of the outmoded end= and err= (not realized so far  action). Note that a 0 value means success, a positive value means an error has occurred, and a negative value means the end of record or end of file was encountered.


Variable / parameter declarations:

· Implicit none: All subroutines and functions must include an "IMPLICIT NONE" statement. Thus all variables must be explicitly typed. It also allows the compiler to detect typographical errors in variable names. For MODULEs, one IMPLICIT NONE statement in the modules definition section is sufficient.
· Data initialization: Improper data initialisation is another common source of errors. A variable could contain an initial value you did not expect. This can happen for several reasons, e.g. the variable has never been assigned a value, its value is outdated, memory has been allocated for a pointer but you have forgotten to initialise the variable pointed to. Some compilers initialise variables to zero but when you port your code to another computer that does not do this previously working code will no longer work this can take some time to diagnose and longer to resolve. To avoid such mishaps initialise variables as close as possible to where they are first used. If possible, give a default initial value in the declaration statement. 
· Private attribute: Modules variables and routines should be encapsulated by using the PRIVATE attribute. What shall be used outside the module can be declared PUBLIC instead. Use USE with the ONLY attribute to specify which of the variables, type definitions etc. defined in a module are to be made available to the using routine. Of course you do not need to add the ONLY attribute if you include the complete module or almost all of its public declarations.
· Constants and magic numbers: Magic numbers should be avoided. Physical constants (e.g. pi, gas constants) must never be hardwired into the executable portion of a code. Instead, a mnemonically named variable or parameter should be set to the appropriate value, probably in the setup routine for the package. We realize that many parameterizations rely on empirically derived constants or fudge factors, which are not easy to name. In these cases it is not forbidden to leave such factors coded as magic numbers buried in executable code, but comments should be included referring to the source of the empirical formula. Hard-coded numbers should never be passed through argument lists. One good reason for this rule is that a compiler flag, which defines a default precision for constants, cannot be guaranteed. Fortran 2003 allows specification of the precision of constants through the "_" compile-time operator (e.g. 3.14_dp or 365_i8). So if you insist on passing a constant through an argument list, you must also include a precision specification in the calling routine. If this is not done, a called routine that declares the resulting dummy argument as, say, real(dp) or 8 bytes, will produce erroneous results if the default floating point precision is 4 byte.
· Interface to C: (description to be done)
· Parameter declaration: Variables used as constants should be declared with attribute PARAMETER and used always without copying to local variables. This prevents from using different values for the same constant or changing them accidentally.
· Usage of the DIMENSION statement or attribute is required in declaration statements.
· The “::” notation is quite useful to show that this program unit declaration part is written in standard FORTRAN syntax, even if there are no attributes to clarify the declaration section. Always use the notation <blank>::<two blanks> to improve readability.
· Declare the length of a character variable using the CHARACTER (len=xxx) syntax - the len specifier is important because it is possible to have several kinds for characters (e.g. Unicode using two bytes per character, or there might be a different kind for Japanese e.g.. NEC).
· Never use a variable name identical to a FORTRAN keyword (the same holds for subroutine names).
· Variable/parameter description: Every single variable/parameter declaration must be accompanied with a comment field expressed with “!<” characters followed by the comment text all on the same line as the declaration itself.
For example:

REAL(wp) ::  c_surface = 20000.0_wp  !< Surface (skin) heat capacity

This allows an easy search of where and how a variable is declared using the unix command: “grep var *90 |grep !<”.
In case of very long declaration statements, add the description in the following line(s) preceded by “!<”, and connect the description to the variable declaration by adding “!<” at the end of the declaration line:

REAL(wp), SAVE, DIMENSION(:,:,:), ALLOCATABLE, TARGET ::  long_line_var !<
     !< description in following line, and in case of very long
     !< description (only if really necessary) use additional lines   

 (include explanation of usage by doxygen here)
· Precision: Parameters and variables should not rely on vendor-supplied flags to supply a default floating point precision or integer size. The F2003 KIND feature should be used instead. Always use the default REAL- (wp) and INTEGER- (iwp) working precision in the respective declaration statements:

REAL(wp)     ::  vara
INTEGER(iwp) ::  counter

The working precisions are set in module file mod_kinds. The defaults are 8 Byte REAL (double precision) and 4 Byte INTEGER (single precision). Only set distinct precisions if they are absolutely required, e.g. for INTEGER counters which may become larger than 232-1 or for REAL quantities which are required in single precision:

INTEGER(idp) ::  particle_id   !< particle counter
REAL(sp), DIMENSION(:,:,:), ALLOCATABLE ::  lpf  !< array for NetCDF output

· Intent: All dummy arguments must include the INTENT clause in their declaration. (still not realized throughout the code  action) This is extremely valuable to someone reading the code, and can be checked by compilers. A common mistake is to put the wrong type of variable in a routine call. So, develop the habit of checking types of arguments in parameter lists. Many modern compilers, especially for Fortran 2003, check for consistent use within a file or across files using inter-procedural analysis. Compilers for Fortran 2003 will also flag up errors at link time if there are explicit or implicit interfaces.
· Interface blocks: Explicit interface blocks are required between routines if optional or keyword arguments are to be used. They also allow the compiler to check that the type, shape and number of arguments specified in the CALL are the same as those specified in the subprogram itself. Fortran 2003 compilers can automatically provide explicit interface blocks for routines contained in a module.






Content rules

Layout
There is a considerable difference that can be made in the readability and the correctness of the interpretation of a piece of code just by its layout.
An enormous difference can be made purely by the use of white space and the judicious use of brackets. A good visual layout shows the logical structure of a program. Typically, programmers use indentation and other white space to show the logical structure:

· White spaces allow you to ensure that related statements are grouped together.
· Unrelated statements can be separated from each other by inserting blank lines. You can also use blank lines to divide groups of related statements into paragraphs, to separate routines from each other, and to highlight comments.
· Align elements that belong together, such as equal signs in groups of related assignments. Do not align elements if they do not belong together, that is only misleading. Align data declarations. Use only one declaration line for each single variable. It is easier to put a comment next to each declaration and modify each declaration accordingly as it is self-contained. Avoid aligning similar items under the same type name (e.g. gathering all integers under one integer-declaration).
· You can use declarations to judiciously group together connected variables; You should indent statements under those statement to which they are logically subordinate.
· Indent a comment to lie with its corresponding code. Otherwise, you will break up the logical structure which was meant to be stressed by indentation. Set off each comment with at least one blank line. Also avoid extremely long comments that will be wrapped when printed as these too will clutter the program flow. Do no be afraid to spread a comment over multiple lines.
· For complicated expressions, put separate conditions on separate lines. Do not be afraid to use brackets to remove any ambiguity. Be as explicit as you can.
· Separate each routine from other routines in the same file or module with at least two blank lines to produce a visual difference between single blank lines that are part of a routine.

Sometimes a statement has to be broken across lines, e.g. because it is longer than programming standards allow. Try to keep things together that belong together, such as array references and arguments to a routine. Do not mix declarations of shared variables (PUBLIC) with declarations of local variables.
Avoid over emphasising comments within modules. If every routine and comment is marked with a line of asterisks, so many things are highlighted that nothing is really clear. In addition, they are more difficult to maintain.

· Variable comments: Variables will be declared 1 per line, with a comment field expressed with "!<" charactera followed by the comment text all on the same line as the declaration. Multiple comment lines describing a single variable are acceptable when necessary. Variables of a like function and type may be grouped together on a single line. For example:

INTEGER ::  i, j, k   !< dummy loop indices

· Commenting style: comments should be written in the form:

!
!-- Describe what is going on and if it is too long use another ‘!--’ for
!-- proper alignment

Short comments may be included on the same line as executable code using the "!" characters preceded by at least three blank space and followed by one blank space and the description

zx = zx * zzy   ! Describe what is going on 

but these short comments should be avoided in general.

Key features of this style are 1) it always starts with a "!" in the first column required for proper automatic indentation, 2) The text starts at the column given by the respective code indentation, and 3) the text may be offset above by a blank line, e.g.:

          x = SQRT( a**2 + b**2 )

!
!--       Describe what is going on in the loop
          DO  i = 1, n
!
!--          Describe the meaning of the IF-block
             IF ( MOD( i, 2 ) == 0 )  THEN

· FORTRAN keywords and intrinsic function or routine: FORTRAN keywords and intrinsic function or routine are written in upper case, the remaining code in lower case. This participates to a better readability of a code. Use ENDDO instead of END DO, ELSEIF instead of ELSE IF, and ENDIF instead of END IF statements. Subroutine, function, parameter, and variable names from external libraries / modules (e.g. MPI, FFTW or NetCDF libraries) are always written in upper case.

· Loops: Loops should be structured with the DO - ENDDO construct as opposed to numbered loops. In case of a long loop, a self-descriptive label can be used (i.e. not just a number!).

· Listings: Members of listings should be given in alphabetical order, for example:

USE arrays_3d,                                                             &
    ONLY:  pt_init, q_init, ref_state, sa_init, ug, u_init, v_init, vg

NAMELIST /d3par/  averaging_interval, averaging_interval_pr,               &
                  cpu_log_barrierwait, create_disturbances,                &
                  cross_profiles, cross_ts_uymax, cross_ts_uymin,          &
                  data_output, data_output_masks

· Continuation lines: Note that the Fortran 2003 standard defines a limit of 39 continuation lines. Code lines, which are continuation lines of assignment statements, should begin to the right of the column of the assignment operator, for example:

arr1 = arr2 + arr3 * arr4 + ... &
     + arr7 * arr8 + arr9 * arr10

Similarly, continuation lines of subroutine calls and dummy argument lists of subroutine declarations must have the arguments aligned to the right of the "(" character. Examples of each of these constructs are:

CALL sub76( px, py, pz, pw, pa, &
            pb, pc, pd, pe )

SUBROUTINE sub76( px, py, pz, pw, pa, &
                  pb, pc, pd, pe )
END SUBROUTINE sub76

In case of more than one continuation line, the continuation marks (ampersand character &) should be vertically aligned, preferably in column 80:

NAMELIST /d3par/  averaging_interval, averaging_interval_pr,               &
                  cpu_log_barrierwait, create_disturbances,                &
                  cross_profiles, cross_ts_uymax, cross_ts_uymin           &

· Indentation: Code as well as comment lines within loops, if-blocks, continuation lines, MODULE or SUBROUTINE statements will be indented 3 characters for readability. (except for CONTAINS that remains at first character)

MODULE mod1
   REAL(wp) ::  xx
CONTAINS
   SUBROUTINE sub76( px, py, pz, pw, pa, &
                     pb, pc, pd, pe )
      <instruction>
   END SUBROUTINE sub76
END MODULE mod1

· Spacing conventions

· Use blank space, in the horizontal and vertical, to improve readability. In particular try to align related code into columns (especially in case of complex equations which extend over several continuation lines). For example, instead of:

!
!-- Initialize Variables
    zx=1.e0_wp
    meaningful_name=3.e0_wp
    SillyName=2.e0_wp

write:

!
!-- Initialize variables
    zx              = 1.0E0_wp
    meaningful_name = 3.0E0_wp
    silly_name      = 2.0E0_wp

In formulas which extend over one or more continuation lines, try to align related parts as well as relating pairs of opening and closing brackets:

    f_mg(kc,jc,ic) =  1.0_wp / 64.0_wp * (                          &
                      8.0_wp * r(k,j,i)                             &
                   +  4.0_wp * ( r(k,j,i-1)     + r(k,j,i+1)     +  &
                                 r(k,j+1,i)     + r(k,j-1,i)     )  &
                   +  2.0_wp * ( r(k,j-1,i-1)   + r(k,j+1,i-1)   +  &
                                 r(k,j-1,i+1)   + r(k,j+1,i+1)   )  &
                   + 16.0_wp * r(k,j,i)                             &
                   +  4.0_wp * r(kp1,j,i)                           &
                   +  2.0_wp * ( r(kp1,j,i-1)   + r(kp1,j,i+1)   +  &
                                 r(kp1,j+1,i)   + r(kp1,j-1,i)   )  &
                   +           ( r(kp1,j-1,i-1) + r(kp1,j+1,i-1) +  &
                                 r(kp1,j-1,i+1) + r(kp1,j+1,i+1) )  &
                                          )

· Use a blank space after a comma (i.e. "a, b, c") except for the indices of an array.
· Use a blank space before and after all operators (+, -, *, =, /), including logical operators (.AND., .OR., ==, /=, etc.), but except the ** operator, where no blank space is used at all. Blank space is not used for oerators within expressions for array indices. 
· Use only one blank space put on the interior side of a bracket, except for brackets related to arrays where no space is needed. There may be more than one blank space in order to get better alignment in case of long formulas with several continuation lines.
· Do not use tab characters (an extension!) in your code: this will ensure that the code looks as intended when ported.
· Formatted I/O: Avoid separating the information to be output from the formatting information on how to output it on I/O statements. (still needs to be done)
· Array arguments: Do not implicitly change the shape of an array when passing it into a subroutine. Although actually forbidden in the FORTRAN77 standard it was very common practice to pass n dimensional arrays into a subroutine where they would, say, be treated as a one dimensional array. Though officially banned in Fortran 95, this practice is still possible with external routines for which no interface block is supplied. The danger of this method is that it makes certain assumptions about how the data is stored.
· Operators: Use of the operators <, >, <=, >=, ==, /= is strongly recommended instead of their deprecated counterparts, .LT., .GT., .LE., .GE., .EQ., and .NE. The motivation is readability. In general use the notation: <Blank><Operator><Blank>
· Array syntax: Array notation should be used whenever possible. This should help optimization regardless what machine architecture is used (at least in theory) and will reduce the number of lines of code required. To improve readability the array shape must be shown in brackets (still needs to be done throughout the code), e.g.: 

onedarraya(:)  = onedarrayb(:) + onedarrayc(:)
twodarray(:,:) = scalar * anothertwodarray(:,:)

When accessing sections of arrays, for example in finite difference equations, do so by using the triplet notation on the full array, e.g.:

twodarray(:,2:len2) = scalar                                          &
                    * ( twodarray2(:,1:len2-1 )                       &
                      - twodarray2(:,2:len2   ) )

Note: In order to improve readability of long, complicated loops, explicitly indexed loops should be preferred. In general when using this syntax, the order of the loop indices should reflect the following scheme: (best usage of data locality).

DO  i = nxl, nxr
   DO  j = nys, nys
      DO  k = nzb, nzt
         array(k,j,i) = ...
      ENDDO
   ENDDO
ENDDO

· Spacing conventions for control structures. Blank spaces are shaded in the following examples.

IF structures:


IF ( a < b  .AND.  ( c == 1  .OR.  exception ) )  THEN
   b = a
ELSEIF ( a == b )  THEN
   b = c
ENDIF

In case that conditions given in the IF or ELSEIF clause require continuation lines, the corresponding THEN should be aligned with the IF:

IF ( a < b  .AND.  ( c == 1  .OR.  exception ) .AND.   &
     second_condition  .AND. third_condition )         &
THEN
   b = a
ELSEIF ( a == b )  THEN
   b = c
ENDIF

CASE structures:

SELECT CASE ( TRIM( timestep_scheme ) )
.  
   CASE ( 'euler' )
      intermediate_timestep_count_max = 1
.
   CASE DEFAULT
      message_string = 'unknown timestep scheme'
.
END SELECT

DO-loop structures:

        a = before_loop

 label: DO  k = nzb_u_inner(j,i)+1, nzt
           dzu(k) = zu(k+1) – zu(k)
        ENDDO label

· Two blanks after closing brackets of conditions:

WHERE ( rif1d < zeta_min )  rif1d = zeta_min
IF ( a < b )  a = b

· Miscellaneous:

WRITE ( io, ‘(F5.4)’ )  vara

USE indices,     &
    ONLY:  nzb, nzt

a = SQRT( a**2 + MAX( b, c )**2 )


Doxygen rules

(to be written by Helge)


Header

Every PALM FORTRAN file requires a doxygen statement with the respective filename as the first line, followed by the Gnu Public License (GPL) agreement, followed by revision comments, followed by a short description of the module/routine. The “!>” characters in the description part make the comments available for the doxygen system. An example:

!> @file message.f90
!--------------------------------------------------------------------------------!
! This file is part of PALM.
!
! PALM is free software: you can redistribute it and/or modify it under the terms
! of the GNU General Public License as published by the Free Software Foundation,
! either version 3 of the License, or (at your option) any later version.
!
! PALM is distributed in the hope that it will be useful, but WITHOUT ANY
! WARRANTY; without even the implied warranty of MERCHANTABILITY or FITNESS FOR
! A PARTICULAR PURPOSE.  See the GNU General Public License for more details.
!
! You should have received a copy of the GNU General Public License along with
! PALM. If not, see <http://www.gnu.org/licenses/>.
!
! Copyright 1997-2014 Leibniz Universitaet Hannover
!--------------------------------------------------------------------------------!
!
! Current revisions:
! -----------------
! 
! 
! Former revisions:
! -----------------
! $Id: message.f90 1683 2015-10-07 23:57:51Z knoop $
!
! 1682 2015-10-07 23:56:08Z knoop
! Code annotations made doxygen readable 
!
! 213 2008-11-13 10:26:18Z raasch
! Initial revision
!
! Description:
! ------------
!> Handling of the different kinds of messages.
!> Meaning of formal parameters:
!> requested_action: 0 - continue, 1 - abort by stop, 2 - abort by mpi_abort
!> message_level: 0 - informative, 1 - warning, 2 - error
!> output_on_pe: -1 - all, else - output on specified PE
!> file_id: 6 - stdout (*)
!> flush: 0 - no action, 1 - flush the respective output buffer
!------------------------------------------------------------------------------!
 SUBROUTINE message( routine_name, message_identifier, requested_action, &
                     message_level, output_on_pe, file_id, flush )

In order to use the “keyword substitution” feature of subversion, a newly created file must contain the Id-keyword

! Former revisions:
! -----------------
! $Id$

After the first commit, subversion will automatically substitute the “Id” string with the current timestamp:

! Former revisions:
! -----------------
! $Id: message.f90 1683 2015-10-07 23:57:51Z raasch $

Substitution of the Id-keyword has to be activated with svn-command

      svn propset svn:keywords "Id" message.f90


Revision History

Any code modification should immediately be recorded under “current revisions”, e.g.

!
! Current revisions:
! -----------------
! bugfix for outflow Neumann boundary conditions at bottom and top,
! boundary conditions for two new prognostic equations (nr, qr) of the 
! two-moment cloud scheme added
! 
! Former revisions:
! -----------------
! $Id: boundary_conds.f90 1718 2015-11-11 16:29:35Z raasch $
!
! 1717 2015-11-11 15:09:47Z raasch
! Bugfix: index error in outflow conditions for left boundary

After the code has been checked and verified, it can be submitted to the subversion repository using the svn-command svn commit –m “short description of revisions” trunk . This will change the timestamp in the $Id line:

! two-moment cloud scheme added
! 
! Former revisions:
! -----------------
! $Id: boundary_conds.f90 1737 2015-11-15 09:11:53Z raasch $

Take this timestamp and the current revision comment lines and create a new entry under former revisions:

!
! Current revisions:
! -----------------
! 
! 
! Former revisions:
! -----------------
! $Id: boundary_conds.f90 1737 2015-11-15 09:11:53Z raasch $
!
! 1737 2015-11-15 09:11:53Z raasch
! bugfix for outflow Neumann boundary conditions at bottom and top,
! boundary conditions for two new prognostic equations (nr, qr) of the 
! two-moment cloud scheme added
!
! 1717 2015-11-11 15:09:47Z raasch
! Bugfix: index error in outflow conditions for left boundary

You can also use the script “document_changes” to move the timestamps automatically. The script will process all files which have entries under “Current revisions”.

Check-in the modified file again with svn commit –m “last commit documented” trunk . Of course, this will change the timestamp again:


! Current revisions:
! -----------------
! 
! 
! Former revisions:
! -----------------
! $Id: boundary_conds.f90 1738 2015-11-15 09:20:14Z raasch $
!
! 1737 2015-11-15 09:11:53Z raasch


General optimization rules
to be done

